
RÉSUMÉ

Écrite en langage MAPLE, une simulation des phénomènes de diffraction à l’infini
par une ouverture, rectangulaire ou circulaire, est présentée. On simule aussi le rôle
joué par la diffraction liée à la largeur de deux fentes d’Young dans la figure d’interfé-
rences qu’elle produit à l’infini.

INTRODUCTION

Dans les deux précédentes communications, on a présenté un programme de simu-
lation en trois dimensions, d’interférences non localisées produites par des sources lumi-
neuses ponctuelles synchrones. Lorsqu’on se limite à l’observation à l’infini, la projec-
tion en perspective ne s’impose plus et l’écriture des programmes de calcul formel s’en
trouve simplifiée. Dans ce qui suit, on décrit et commente trois programmes écrits en
langage MAPLE, destinés à simuler :

– La diffraction à l’infini d’une onde plane monochromatique traversant sous inci-
dence normale une ouverture rectangulaire dont la longueur verticale (notée H dans le
listing) est le double de la largeur horizontale (notée L dans le listing).

– La diffraction à l’infini d’une onde plane monochromatique traversant sous inci-
dence normale une ouverture circulaire de diamètre connu (noté a dans le listing).

– Le couplage interférences-diffraction dû à l’influence de la largeur (L) de deux fentes
d’Young infiniment longues d’entraxe donné (noté a dans le listing), éclairées sous in-
cidence normale par une onde plane monochromatique. On se limite au cas d’une obser-
vation à l’infini.

ORGANISATION DES PROGRAMMES

Options prises pour le repérage dans l’espace géométrique

– L’axe Ox du repère orthonormé cartésien Oxyz servant aux calculs contient le centre
C du diaphragme.

– Le plan de l’ouverture est parallèle au plan Oyz et la direction de l’onde incidente est
celle de l’axe Ox. La plus grande dimension de la fente rectangulaire ou des fentes
d’Young est parallèle à l’axe Oz.

– α représente l’écart angulaire du rayon diffracté par rapport au plan xOz (noté alfa
dans les listings).

– β représente son écart angulaire par rapport au plan xOy (noté beta dans les listings).

Données initiales à introduire au choix de l’utilisateur

– Les dimensions géométriques du diaphragme (L et H pour l’ouverture rectangulaire,
a pour le trou circulaire, a et L pour les fentes).

– L’étendue du champ d’observation.

– La longueur d’ondeλ de la lumière émise par les sources (notée lambda dans les lis-
tings).

– Le maillage de la grille du tracé, qui conditionne sa résolution.

Déroulement des calculs

Les algorithmes choisis sont sommairement explicités dans les commentaires ac-
compagnant les trois listings figurant en fin de texte (voir Annexe). Les formules utili-
sées pour les calculs d’éclairement relatif sont celles de la théorie physique, avec
l’usage de la fonction de BESSELd’ordre un dans le cas d’une ouverture circulaire (les
fonctions de BESSELfont partie du noyau ou de la bibliothèque standard de MAPLE).

Résultats

Voir les figures 1 à 6,avec un affichage vidéo monochrome en noir et blanc, obtenu
en donnant la même valeur aux trois variables R (rouge), G (vert), B (bleu) de l’outil
«color».

COMMENTAIRES

A propos des algorithmes

Pour un choix de maillage de grille de (150× 150) points, le temps de calcul est
bien plus long pour le diaphragme circulaire que pour le diaphragme rectangulaire. On
peut gagner du temps et aussi de la finesse pour les fentes d’Young en choisissant un
maillage très fin pourα (1000) et très grossier pour z (2) car le phénomène observé ne
dépend que deα , puisque les fentes sont choisies infiniment longues.

A propos des résultats

L’observation sur l’écran d’un moniteur vidéo de la figure issue des calculs est
plutôt décevante. On ne voit qu’une partie de la tache centrale de diffraction et les
débordement latéraux (annulaires ou cruciformes) passent inaperçus. Le rendu est
meilleur sur papier, à l’aide d’une imprimante, mais cela augmente beaucoup la durée
de travail.

Les figures 3 à 6montrent bien que l’étendue du champ d’interférences observa-
bles est fortement conditionnée par la largeur des fentes.

Figure 2 : Diffraction à l’infini par un trou circulaire
Données initiales : celles du listing du programme

diamètre du trou = largeur de la fente de la figure 1, graduations en radian
Grille choisie pour le dessin : (150× 150) points.

Figure 1 : Diffraction à l’infini par une fente rectangulaire
Données initiales : celles du listing du programme

hauteur de la fente : double de sa largeur alpha et beta en radian
Grille choisie pour le dessin : (150× 150) points.

Figure 4 : Influence de la largeur des fentes d’Young
Données initiales : celles du listing du programme

sauf : largeur = entraxe/16
Grille choisie pour le dessin : (1000× 2) points.

Figure 3 : Influence de la largeur des fentes d’Young
Données initiales : celles du listing du programme

largeur = entraxe/32
Grille choisie pour le dessin : (1000× 2) points.

Figure 6 : Influence de la largeur des fentes d’Young
Données initiales : celles du listing du programme

sauf : largeur = entraxe/4
Grille choisie pour le dessin : (1000× 2) points.

Figure 5 : Influence de la largeur des fentes d’Young
Données initiales : celles du listing du programme

sauf : largeur = entraxe/8
Grille choisie pour le dessin : (1000× 2) points.



Annexe

Les commentaires sont en italique

PROGRAMME POUR DIFFRACTION A L’INFINI PAR UN DIAPHRAGME RECTANGULAIRE

> with(plots,display):

Données initiales au choix de l’utilisateur

> L:=1: # largeur fente diffractante

> H:=2*L: # hauteur fente diffractante

> lambda:=500E-6*L: # longueur d’onde incidente

> Grille:=150: # réglage finesse maillage du dessin

Calculs préliminaires

> k:=2*Pi/lambda: # norme vecteur d’onde

> Bord:=3*lambda/L: # champ angulaire pour limites du tracé

Calculs pour le tracé de la tache de diffraction

> sinc:=u→ sin(u)/(u): # définition de la fonction sinus cardinal

> EclR:=(sinc(k*L*alfa/2)*sinc(k*H*beta/2))^2: # éclairement relatif en M

> Tache:=plot3d([0,alfa,beta],alfa=-Bord..Bord,beta=-Bord..Bord,
grid=[Grille,Grille],color=COLOR(RGB,EclR,EclR,EclR),
orientation=[0,90],style=patchnogrid,axes=boxed,
tickmarks=[DEFAULT,3,3],labels=[‘x’,’a’,’b’],labelfont=[SYMBOL,16]):

Dessin

> display(Tache);

PROGRAMME POUR DIFFRACTION A L’INFINI
PAR UN DIAPHRAGME CIRCULAIRE

> with(plots,display):

Données initiales au choix de l’utilisateur

> a:=1: # diamètre du trou

> lambda:=500E-6*a: # longueur d’onde incidente

> Grille:=150: # réglage finesse maillage du dessin

Calculs préliminaires

> k:=2*Pi/lambda: # norme vecteur d’onde

> Bord:=3*lambda/L: # champ angulaire pour limites du tracé

Calculs pour le tracé de la tache de diffraction

> readlib(BesselJ):
# chargement de la bibliothèque BESSEL interne
à MAPLE.

> ADiff:=alfa → (BesselJ(1,k*a*alfa/2)/
(k*a*alfa/2)):

# amplitude diffractée

> EclR:=(ADiff(alfa)/ADiff(1E-6))^2: # éclairement relatif en M

> Tache:=plot3d([alfa,theta,0],alfa=1E-6..Bord,theta=0..2*Pi,coords=cylindrical,
grid=[Grille,Grille],color=COLOR(RGB,EclR,EclR,EclR),
orientation=[0,180],style=patchnogrid,axes=boxed,
tickmarks=[DEFAULT,3,3]):

Dessin

> display(Tache);

PROGRAMME POUR COUPLAGE DIFFRACTION - INTERFERENCES A L’INFINI
DANS L’EXPERIENCE D’YOUNG

> with(plots,display):

Données initiales au choix de l’utilisateur

> a:=1: # entraxe entre les 2 fentes sources

> L:=a/32: #largeur relative de chaque fente

> lambda:=500E-6*a: # longueur d’onde incidente

> NbF:=21:
# nombre de franges dans le champ angulaire
latéral

> Grille:=1000: # réglage finesse maillage du dessin

Calculs préliminaires

> Champ:=NbF*lambda*a/2: # étendue champ angulaire de tracé des franges

> k:=2*Pi/lambda: # norme vecteur d’onde

Calculs pour le tracé de la figure d’interférences

> sinc:=u→ sin(u)/u: # définition fonction sinus cardinal

> AmpliDiffr:=sinc(k*L*alfa/2): # amplitude de diffraction

> EclR:=(1+cos(k*a*alfa))*AmpliDiffr^2/2: # éclairement relatif total

> Franges:=plot3d([0,alfa,z],alfa=-Champ..Champ,z=-Champ/2..Champ/2,
grid=[Grille,2],color=COLOR(RGB,EclR,EclR,EclR),
orientation=[0,90],style=patchnogrid,scaling=constrained,
axes=none):

Dessin

> display(Franges);

